IVRG EXAMPLE
$g = 4$, $G = \langle 72, 40\rangle$

DAVID SWINARSKI

The group $G = \langle 72, 40\rangle$ is the automorphism group of a genus 4 curve [2]. We use `DecomposeGAction` to find equations for two curves with this automorphism group. One is smooth, the other is singular.

First, we compute the center of G in `Magma`.

```
> SmallGroupDatabase();
> H:=SmallGroup(96,64);
> Center(H);
GrpPC of order 1
PC-Relations:
```

It is trivial. Hence, the curve C is nonhyperelliptic, and we seek one quadric and one cubic in four variables.

Next, we find matrix generators for the action of $\mathrm{Aut}(C)$ on the vector space $H^0(C, K)$. These are given in [1], Prop. 2.4.f.2, p. 288.

Let $w = e^{2\pi i/3}$. Then the generators are

$$\begin{pmatrix} 1 & 0 & 0 & 0 \\ 0 & 1 & 0 & 0 \\ 0 & 0 & w & 0 \\ 0 & 0 & 0 & w^2 \end{pmatrix}, \begin{pmatrix} 0 & 1 & 0 & 0 \\ 1 & 0 & 0 & 0 \\ 0 & 0 & -1 & 0 \\ 0 & 0 & 0 & -1 \end{pmatrix}, \begin{pmatrix} 0 & 0 & 0 & 1 \\ 0 & 0 & 1 & 0 \\ 0 & 1 & 0 & 0 \\ 1 & 0 & 0 & 0 \end{pmatrix}$$

which we call A, B, and C in the `Magma` session below.

```
> K<w>:=CyclotomicField(3);
> GL4K:=GeneralLinearGroup(4,K);
> A:=elt<GL4K | 1,0,0,0, 0,1,0,0, 0,0,w,0, 0,0,0,w^2>;
> B:=elt<GL4K | 0,1,0,0, 1,0,0,0, 0,0,-1,0, 0,0,0,-1>;
> C:=elt<GL4K | 0,0,0,1, 0,0,1,0, 0,1,0,0, 1,0,0,0>;
> G:=sub<GL4K | A,B,C>;
> IdentifyGroup(G);
<72,40>
> load "DecomposeGAction.txt";
Loading "DecomposeGAction.txt"
> S<a,b,c,d>:=PolynomialRing(K,4);
> DecomposeGAction(G,S,2);
[
    rec<recformat<CharacterRow, Dimension, Elements> |
        CharacterRow := 1,
        Dimension := 1,
        Elements := [
            a*b + c*d
        ]
        >,
    rec<recformat<CharacterRow, Dimension, Elements> |
```

Date: October 4, 2010.

```
        CharacterRow := 2,
        Dimension := 1,
        Elements := [
            a*b - c*d
        ]
        >,
    rec<recformat<CharacterRow, Dimension, Elements> |
        CharacterRow := 7,
        Dimension := 4,
        Elements := [
            a^2,
            b^2,
            c^2,
            d^2
        ]
        >,
    rec<recformat<CharacterRow, Dimension, Elements> |
        CharacterRow := 8,
        Dimension := 4,
        Elements := [
            a*c,
            a*d,
            b*c,
            b*d
        ]
        >
]

> DecomposeGAction(G,S,3);
[
    rec<recformat<CharacterRow, Dimension, Elements> |
        CharacterRow := 2,
        Dimension := 1,
        Elements := [
            a^3 - b^3 - c^3 + d^3
        ]
        >,
    rec<recformat<CharacterRow, Dimension, Elements> |
        CharacterRow := 3,
        Dimension := 1,
        Elements := [
            a^3 - b^3 + c^3 - d^3
        ]
        >,
    rec<recformat<CharacterRow, Dimension, Elements> |
        CharacterRow := 5,
        Dimension := 2,
        Elements := [
            a^3 + b^3,
            c^3 + d^3
```

```
        ]
        >,
    rec<recformat<CharacterRow, Dimension, Elements> |
        CharacterRow := 6,
        Dimension := 4,
        Elements := [
            a^2*c - b*d^2,
            a^2*d + b*c^2,
            a*c^2 - b^2*d,
            a*d^2 + b^2*c
        ]
        >,
    rec<recformat<CharacterRow, Dimension, Elements> |
        CharacterRow := 7,
        Dimension := 4,
        Elements := [
            a^2*c + b*d^2,
            a^2*d - b*c^2,
            a*c^2 + b^2*d,
            a*d^2 - b^2*c
        ]
        >
]
```

This gives us four pairs of polynomials to study: $ab + cd$ or $ab - cd$ with $a^3 - b^3 + c^3 - d^3$ or $a^3 - b^3 - c^3 + d^3$. Actually, we can see that two of these pairs are the same as the other two pairs under the map which permutes c and d, and so we really only need to study $ab + cd$ with $a^3 - b^3 + c^3 - d^3$ and $ab - cd$ with $a^3 - b^3 + c^3 - d^3$.

We study the schemes given by these pairs of polynomials:

```
> P3<a,b,c,d>:=ProjectiveSpace(K,3);
> X1:=Scheme(P3,[a*b+c*d,a^3-b^3+c^3-d^3]);
> IsNonsingular(X1);
true
> X2:=Scheme(P3,[a*b-c*d,a^3-b^3+c^3-d^3]);
> IsNonsingular(X2);
false
```

We see that we get one smooth curve and one singular curve.

We verify that the smooth curve is the curve we are looking for:

```
> Dimension(X1);
1
> X1:=Curve(X1);
> Genus(X1);
4
> AutX1:=AutomorphismGroup(X1);
> Order(AutX1);
72
> PermAutX1:=PermutationRepresentation(AutX1);
> IdentifyGroup(PermAutX1);
<72, 40>
```

Next, we study the singular curve X_2:

```
> IrreducibleComponents(X2);
[
    Scheme over K defined by
    a + w*c,
    b + (-w - 1)*d,
    Scheme over K defined by
    a + (-w - 1)*c,
    b + w*d,
    Scheme over K defined by
    a + c,
    b + d,
    Scheme over K defined by
    a + (w + 1)*d,
    b - w*c,
    Scheme over K defined by
    a - w*d,
    b + (w + 1)*c,
    Scheme over K defined by
    a - d,
    b - c
]
> L:=IrreducibleComponents(X2);
> [ IsNonsingular(Z) : Z in L];
[ true, true, true, true, true, true ]
> [ ArithmeticGenus(Z) : Z in L];
[ 0, 0, 0, 0, 0, 0 ]
> IsEmpty(Intersection(L[1],L[2]));
true
> IsEmpty(Intersection(L[1],L[3]));
true
> IsEmpty(Intersection(L[1],L[4]));
false
> IsEmpty(Intersection(L[1],L[5]));
false
> IsEmpty(Intersection(L[1],L[6]));
false
> IsEmpty(Intersection(L[2],L[3]));
true
> IsEmpty(Intersection(L[2],L[4]));
false
> IsEmpty(Intersection(L[2],L[5]));
false
> IsEmpty(Intersection(L[2],L[6]));
false
> IsEmpty(Intersection(L[3],L[4]));
false
> IsEmpty(Intersection(L[3],L[5]));
false
> IsEmpty(Intersection(L[3],L[6]));
false
```

```
> IsEmpty(Intersection(L[4],L[5]));
true
> IsEmpty(Intersection(L[4],L[6]));
true
> IsEmpty(Intersection(L[5],L[6]));
true
> IsReduced(Intersection(L[1],L[4]));
true
> IsReduced(Intersection(L[1],L[5]));
true
> IsReduced(Intersection(L[1],L[6]));
true
> IsReduced(Intersection(L[2],L[4]));
true
> IsReduced(Intersection(L[2],L[5]));
true
> IsReduced(Intersection(L[2],L[6]));
true
> IsReduced(Intersection(L[3],L[4]));
true
> IsReduced(Intersection(L[3],L[5]));
true
> IsReduced(Intersection(L[3],L[6]));
true
> HilbertPolynomial(Ideal(Intersection(L[1],L[4])));
1
0
> HilbertPolynomial(Ideal(Intersection(L[1],L[5])));
1
0
> HilbertPolynomial(Ideal(Intersection(L[1],L[6])));
1
0
> HilbertPolynomial(Ideal(Intersection(L[2],L[4])));
1
0
> HilbertPolynomial(Ideal(Intersection(L[2],L[5])));
1
0
> HilbertPolynomial(Ideal(Intersection(L[2],L[6])));
1
0
> HilbertPolynomial(Ideal(Intersection(L[3],L[4])));
1
0
> HilbertPolynomial(Ideal(Intersection(L[3],L[5])));
1
0
> HilbertPolynomial(Ideal(Intersection(L[3],L[6])));
1
```

0

From the results above, we conclude that the singular curve X_2 has 6 irreducible components, each of which is the Riemann sphere. If we call these Z_1 through Z_6, we find that $Z_i \cap Z_j \neq \emptyset$ if and only if $i \in \{1, 2, 3\}$ and $j \in \{4, 5, 6\}$. When $Z_i \cap Z_j \neq \emptyset$, we see that $Z_i \cap Z_j$ is a single point. Thus, we can describe the arrangement of the components Z_1 through Z_6 using the graph $K_{3,3}$.

References

[1] Izumi Kuribayashi and Akikazu Kuribayashi, *Automorphism groups of compact Riemann surfaces of genera three and four*, J. Pure Appl. Algebra **65** (1990), no. 3, 277–292, DOI 10.1016/0022-4049(90)90107-S. MR1072285 (92a:30041) ←1

[2] K. Magaard, T. Shaska, S. Shpectorov, and H. Völklein, *The locus of curves with prescribed automorphism group*, Sūrikaisekikenkyūsho Kōkyūroku **1267** (2002), 112–141, available at arXiv:math.AG/0205314. Communications in arithmetic fundamental groups (Kyoto, 1999/2001). MR1954371 ←1

Software Packages Referenced

[3] School of Mathematics and Statistics Computational Algebra Research Group University of Sydney, MAGMA *computational algebra system* (2008), available at http://magma.maths.usyd.edu.au/magma/. Version 2.15-1. ←